

# Le langage *Vier*

pour le cours de construction de compilateurs de 2007

révision  $\gamma_7$

## Grammaire lexicale

```
input ::= { whitespace | comment | token }
token ::= ident
        | number
        | string
        | "true" | "false"
        | '(' | ')' | '{' | '}'
        | '=' | ':' | ';' | '.'
        | "class" | "extends"
        | "field" | "method"
        | "Int" | "Null"
        | "new" | "null" | "this"
        | "if" | "then" | "else"
        | "readInt" | "readChar"
        | "printInt" | "printChar"
        | "return"
        | '+' | '-' | '*' | '/' | '%'
        | "==" | '<' | '>' | "&&" | "||"
comment ::= "//" { cchar }
ident ::= letter { letter | digit | '_' }
number ::= '0' | digit1 { digit }
string ::= '"' { schar } '"'
letter ::= 'a' | ... | 'z' | 'A' | ... | 'Z'
digit ::= '0' | digit1
digit1 ::= '1' | ... | '9'
```

Un caractère est "whitespace" si la méthode "Character.isWhitespace" de la librairie Java l'accepte. Un caractère est "cchar" s'il ne définit pas de nouvelle ligne. Un caractère est "schar" s'il est "cchar" et pas '"'.

Dans la grammaire lexicale, "abc" doit être lu comme 'a' 'b' 'c'.

## Grammaire syntaxique

```

Program ::= { ClassDef } Expr
ClassDef ::= "class" ident [ "extends" ident ] '{' { Member } '}'
Member  ::= FieldDef
        | MethDef
    Type ::= ident | "Int" | "Null"
    Formal ::= ident ':' Type
    Formals ::= [ Formal { ';' Formal } ]
    FieldDef ::= "field" Formal '=' Expr
    MethDef ::= "method" ident '(' Formals ')' ':' Type '=' Expr
    Expr ::= "if" Expr "then" Expr "else" Expr
        | CmpExpr
    CmpOp ::= "==" | '<' | '>'
    CmpExpr ::= [ SumExpr CmpOp ] SumExpr
    SumOp ::= '+' | '-' | "||"
    SumExpr ::= [ SumExpr SumOp ] Term
    ProdOp ::= '*' | '/' | '%' | "&&"
    Term ::= [ Term ProdOp ] Factor
    Factor ::= ident
        | number
        | string
        | "true" | "false"
        | "null" | "this"
        | "new" Type
        | Factor ':' ident
        | Factor '.' '{ Assigns }'
        | Factor '.' ident '(' Exprs ')'
        | '(' Expr ')'
        | '{ Exprs "return" Expr }'
        | "readInt" | "readChar"
        | "printInt" '(' Expr ')'
        | "printChar" '(' Expr ')'
    Exprs ::= [ Expr { ';' Expr } ]
    Assigns ::= ident '=' Expr { ';' ident '=' Expr }

```

## Sucre syntaxique

| Syntaxe concrète                | Syntaxe abstraite                                  |
|---------------------------------|--|
| <code>true</code>               | <code>1</code>                                     |
| <code>false</code>              | <code>0</code>                                     |
| <code>e    f</code>             | <code>¬(¬e ∧ ¬e')</code>                           |
| <code>¬e</code>                 | <code>if e = 0 then 1 else 0</code>                |
| <code>e .{ a = f }</code>       | <code>e.a = f</code>                               |
| <code>e .{ a = f ; ... }</code> | <code>(e.a = f) .{ ... }</code>                    |
| <code>""</code>                 | <code>new Nil</code>                               |
| <code>"a..."</code>             | <code>(new Cons).{ head = a; tail = "..." }</code> |

## Grammaire abstraite

|                  |  |                      |
|------------------|--|----------------------|
| nom référençable | $a, b$   |                      |
| nom de classe    | $t, u$   |                      |
| nombre           | $n$  |                      |
| programme        | $P ::= \overline{D} e$                                     |                      |
| classe           | $D ::= \text{class } t \triangleleft s \{ \overline{d} \}$ |                      |
|                  | $s ::= t$  |                      |
|                  | <b>none</b>  |                      |
| membre           | $d ::= \text{field } a : T = e$                            |                      |
|                  | <b>method</b> $a (\overline{a} : \overline{T}) : T = e$    |                      |
| type             | $T, U ::= t$   |                      |
|                  | <b>Int</b>   |                      |
|                  | <b>Null</b>  |                      |
| expression       | $e, f ::= a$   | paramètre            |
|                  | $n$  | nombre               |
|                  | <b>new</b> $t$   | nouvelle instance    |
|                  | <b>null</b>  | valeur “null”        |
|                  | $e.a$  | sélection de champ   |
|                  | $e.a(\overline{e})$  | appel de méthode     |
|                  | $e.a = e$  | assignation de champ |
|                  | <b>if</b> $e$ <b>then</b> $e$ <b>else</b> $e$              | condition “if”       |
|                  | $e = e$  | comparaison binaire  |
|                  | $e p e$  | op. binaire          |
|                  | <b>readInt</b>   | op. “readInt”        |
|                  | <b>readChar</b>  | op. “readChar”       |
|                  | <b>printInt</b> ( $e$ )                                    | op. “printInt”       |
|                  | <b>printChar</b> ( $e$ )                                   | op. “printChar”      |
|                  | <b>{</b> $\overline{e}$ <b>return</b> $e$ <b>}</b>         | block                |
| opérateur        | $p ::= +   -   *   \div   \%   <   >   \wedge$             |                      |

## Validité du typage

### Symboles et portées

|                       |   |
|-----------------------|---|
| symbole de classe     | $\sigma_c ::= (\overline{t}   \Gamma_f   \Gamma_m)_c$   |
| symbole de champ      | $\sigma_f ::= (T)_f$                                    |
| symbole de méthode    | $\sigma_m ::= (\overline{T}   T)_m$                     |
| symbole de paramètre  | $\sigma_p ::= (T)_p$                                    |
| portée des classes    | $\Gamma_c ::= \overline{t} \mapsto \overline{\sigma}_c$ |
| portée des champs     | $\Gamma_f ::= \overline{a} \mapsto \overline{\sigma}_f$ |
| portée des méthodes   | $\Gamma_m ::= \overline{a} \mapsto \overline{\sigma}_m$ |
| portée des paramètres | $\Gamma_p ::= \overline{a} \mapsto \overline{\sigma}_p$ |

## Règles de typage

Validité d'un programme  $\overline{D} e \diamond$

$$\frac{\Gamma'_c \vdash \overline{D} \Rightarrow \Gamma_c \quad \text{none} \mapsto (\epsilon | \epsilon | \epsilon)_c \vdash \overline{D} \Rightarrow \Gamma'_c \quad \Gamma_c \vdash \overline{D} \diamond \quad \Gamma_c; \epsilon \vdash e : T}{\overline{D} e \diamond}$$

Validité de l'héritage entre classes  $\Gamma_c \vdash D \Rightarrow \Gamma'_c$

$$\frac{s \mapsto (\overline{s} | \epsilon | \epsilon)_c \in \Gamma_c \quad \Gamma'_c = \Gamma_c \uplus t \mapsto (s, \overline{s} | \epsilon | \epsilon)_c}{\Gamma_c \vdash \text{class } t \triangleleft s \{ \overline{d} \} \Rightarrow \Gamma'_c}$$

Validité de la déclaration des membres  $\Gamma_c \vdash D \Rightarrow \Gamma'_c$

$$\frac{s \mapsto (\overline{s} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \quad \Gamma'_f = \Gamma_f + \text{fields}(\overline{d}) \quad \Gamma'_m = \Gamma_m + \text{methods}(\overline{d}) \quad \Gamma'_c = \Gamma_c + t \mapsto (s, \overline{s} | \Gamma'_f | \Gamma'_m)_c}{\Gamma_c \vdash \text{class } t \triangleleft s \{ \overline{d} \} \Rightarrow \Gamma'_c}$$

Validité d'une classe  $\Gamma_c \vdash D \diamond$

$$\frac{\Gamma_c; \text{this} \mapsto (t)_p \vdash \overline{d} \diamond}{\Gamma_c \vdash \text{class } t \triangleleft s \{ \overline{d} \} \diamond}$$

Validité d'un membre  $\Gamma_c; \Gamma_p \vdash d \diamond$

$$\frac{\left. \begin{array}{l} \text{this} \mapsto (t)_p \in \Gamma_p \\ t \mapsto (s, \overline{s} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \\ s \mapsto (\overline{s} | \Gamma'_f | \Gamma'_m)_c \in \Gamma_c \\ a \mapsto (T')_f \in \Gamma'_f \end{array} \right\} \Rightarrow T = T' \quad \Gamma_c \vdash T \diamond \quad \Gamma_c; \Gamma_p \vdash e : U \quad \Gamma_c \vdash U \triangleleft T}{\Gamma_c; \Gamma_p \vdash \text{field } a : T = e \diamond}$$

$$\frac{\left. \begin{array}{l} \text{this} \mapsto (t)_p \in \Gamma_p \\ t \mapsto (s, \overline{s} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \\ s \mapsto (\overline{s} | \Gamma'_f | \Gamma'_m)_c \in \Gamma_c \\ a \mapsto (\overline{T'} | T')_m \in \Gamma'_m \end{array} \right\} \Rightarrow \Gamma_c \vdash T \triangleleft T' \wedge \Gamma_c \vdash \overline{T'} \triangleleft \overline{T} \quad \Gamma_c \vdash T \diamond \quad \Gamma_c \vdash \overline{T} \diamond \quad \Gamma_c; \Gamma_p + \text{params}(\overline{a} : \overline{T}) \vdash e : U \quad \Gamma_c \vdash U \triangleleft T}{\Gamma_c; \Gamma_p \vdash \text{method } a (\overline{a} : \overline{T}) : T = e \diamond}$$

**Validité d'une expression**  $\Gamma_c; \Gamma_p \vdash e : T$

$$\frac{a \mapsto (T)_p \in \Gamma_p}{\Gamma_c; \Gamma_p \vdash a : T} \quad \Gamma_c; \Gamma_p \vdash n : \text{Int} \quad \frac{t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c}{\Gamma_c; \Gamma_p \vdash \text{new } t : t}$$

$$\Gamma_c; \Gamma_p \vdash \text{null} : \text{Null}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : t \quad t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \quad a \mapsto (U)_f \in \Gamma_f}{\Gamma_c; \Gamma_p \vdash e.a : U}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : t \quad t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \quad a \mapsto (\bar{U} | U)_m \in \Gamma_m \quad \Gamma_c; \Gamma_p \vdash \bar{e} : \bar{T} \quad \Gamma_c \vdash \bar{T} \triangleleft \bar{U}}{\Gamma_c; \Gamma_p \vdash e.a(\bar{e}) : U}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : t \quad t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \quad a \mapsto (U)_f \in \Gamma_f \quad \Gamma_c; \Gamma_p \vdash f : T \quad \Gamma_c \vdash T \triangleleft U}{\Gamma_c; \Gamma_p \vdash e.a = f : t}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : \text{Int} \quad \Gamma_c; \Gamma_p \vdash f : U \quad \Gamma_c; \Gamma_p \vdash f' : U' \quad \Gamma_c \vdash U \triangleleft T \triangleright U'}{\Gamma_c; \Gamma_p \vdash \text{if } e \text{ then } f \text{ else } f' : T}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : T \quad \Gamma_c; \Gamma_p \vdash e' : T' \quad \Gamma_c \vdash T \triangleleft T' \vee \Gamma_c \vdash T' \triangleleft T}{\Gamma_c; \Gamma_p \vdash e = e' : \text{Int}}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : \text{Int} \quad \Gamma_c; \Gamma_p \vdash e' : \text{Int}}{\Gamma_c; \Gamma_p \vdash e.p.e' : \text{Int}} \quad \Gamma_c; \Gamma_p \vdash \text{readInt} : \text{Int}$$

$$\Gamma_c; \Gamma_p \vdash \text{readChar} : \text{Int} \quad \frac{\Gamma_c; \Gamma_p \vdash e : \text{Int}}{\Gamma_c; \Gamma_p \vdash \text{printInt}(e) : \text{Int}}$$

$$\frac{\Gamma_c; \Gamma_p \vdash e : \text{Int}}{\Gamma_c; \Gamma_p \vdash \text{printChar}(e) : \text{Int}} \quad \frac{\Gamma_c; \Gamma_p \vdash \bar{e} : \bar{U} \quad \Gamma_c; \Gamma_p \vdash f : T}{\Gamma_c; \Gamma_p \vdash \{ \bar{e} \text{ return } f \} : T}$$

**Validité d'un type**  $\Gamma_c \vdash T \diamond$

$$\frac{t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c}{\Gamma_c \vdash t \diamond} \quad \Gamma_c \vdash \text{Null} \diamond \quad \Gamma_c \vdash \text{Int} \diamond$$

**Sous-typage**  $\Gamma_c \vdash T \triangleleft T$

$$\frac{t \mapsto (\bar{t} | \Gamma_f | \Gamma_m)_c \in \Gamma_c \quad u \in \bar{t} \vee u = t}{\Gamma_c \vdash t \triangleleft u} \quad \Gamma_c \vdash \text{Null} \triangleleft t$$

$$\Gamma_c \vdash \text{Null} \triangleleft \text{Null}$$

$$\Gamma_c \vdash \text{Int} \triangleleft \text{Int}$$

**Borne supérieure**  $\Gamma_c \vdash T \triangleleft T \triangleright T$

$$\frac{\Gamma_c \vdash T \triangleleft U \quad \Gamma_c \vdash T' \triangleleft U \quad \forall U'. \left. \begin{array}{l} \Gamma_c \vdash T \triangleleft U' \\ \Gamma_c \vdash T' \triangleleft U' \end{array} \right\} \Rightarrow \Gamma_c \vdash U \triangleleft U'}{\Gamma_c \vdash T \triangleleft U \triangleright T'}$$

## Notations

Une notation définie sur  $x$  l'est pour tout élément, une notation définie sur  $\Gamma$  l'est pour toute portée.

| Notation  | Interprétation  |
|---|---|
| $\bar{x}$                                       | séquence $x_0, \dots, x_n$ pour $n \in \mathbb{N}$  |
| $x, \bar{x}$                                    | séquence $x, x_0, \dots, x_n$   |
| $\bar{x} \mapsto \bar{\sigma}$                  | portée $x_0 \mapsto \sigma_0, \dots, x_n \mapsto \sigma_n$ pour $n \in \mathbb{N}$  |
| $\epsilon$                                      | portée ou séquence vide   |
| $\Gamma_c \vdash \bar{D} \Rightarrow \Gamma'_c$ | $\left\{ \begin{array}{l} \Gamma_c \vdash D_0 \Rightarrow \Gamma_c^0 \\ \vdots \\ \Gamma_c^{n-1} \vdash D_n \Rightarrow \Gamma'_c \end{array} \right.$  |
| $\Gamma_c \vdash \bar{D} \Rightarrow \Gamma'_c$ | $\left\{ \begin{array}{l} \Gamma_c \vdash D_0 \Rightarrow \Gamma_c^0 \\ \vdots \\ \Gamma_c^{n-1} \vdash D_n \Rightarrow \Gamma'_c \end{array} \right.$  |
| $\Gamma_c \vdash \bar{D} \diamond$              | $\forall D \in \bar{D}. \Gamma_c \vdash D \diamond$   |
| $\Gamma_c; \Gamma_p \vdash \bar{d} \diamond$    | $\forall d \in \bar{d}. \Gamma_c; \Gamma_p \vdash d \diamond$   |
| $\Gamma_c \vdash \bar{T} \diamond$              | $\forall T \in \bar{T}. \Gamma_c \vdash T \diamond$   |
| $\Gamma_c \vdash \bar{T} \triangleleft \bar{U}$ | $\forall T, U \in \bar{T} \times \bar{U}. \Gamma_c \vdash T \triangleleft U$  |
| $\Gamma_c; \Gamma_p \vdash \bar{e} : \bar{T}$   | $\forall e, T \in \bar{e} \times \bar{T}. \Gamma_c; \Gamma_p \vdash e : T$  |
| $\Gamma + x \mapsto \sigma$                     | $\left\{ \begin{array}{ll} \Gamma', x \mapsto \sigma, \Gamma'' & \text{si } \exists \Gamma', \Gamma'', \sigma'. \Gamma = (\Gamma', x \mapsto \sigma', \Gamma'') \\ \Gamma, x \mapsto \sigma & \text{sinon} \end{array} \right.$ |
| $\Gamma \uplus x \mapsto \sigma$                | $\Gamma, x \mapsto \sigma$ si $\forall \Gamma', \Gamma'', \sigma'. \Gamma \neq (\Gamma', x \mapsto \sigma', \Gamma'')$  |
| $fields(\bar{d})$                               | $\biguplus_{\text{field } a:T=e \in \bar{d}} a \mapsto (T)_f$   |
| $methods(\bar{d})$                              | $\biguplus_{\text{method } a(\bar{a}:\bar{T}):T=e \in \bar{d}} a \mapsto (\bar{T}   T)_m$   |
| $params(\bar{a} : \bar{T})$                     | $\bigoplus_{a:T \in \bar{a}:\bar{T}} a \mapsto (T)_p$   |

## Etat des changements

- révision**  $\gamma_7$  Les éléments du symbole de paramètre dans la règle de validité d'une méthode sont dans le bon ordre.
- révision**  $\gamma_6$  La règle de sous-typage entre classes définit correctement une classe comme sous-type d'elle-même.
- révision**  $\gamma_5$  Le sucre syntaxique pour la négation ( $\neg e$ ) est correct.
- révision**  $\gamma_4$  La définition de la borne supérieure est plus simple. Les règles de validité d'un membre sont correctes aussi lorsque le membre défini n'est pas hérité.
- révision**  $\gamma_3$  La grammaire abstraite est syntaxiquement plus proche de la grammaire concrète. Le type "Nothing" s'appelle "Null".
- révision**  $\gamma_2$  La liste des classes parentes du symbole d'une classe ne contient plus la classe elle-même. Le sous-typage entre "Nothing" et "Nothing" est correct.
- révision**  $\gamma_1$  Ajoute les règles de validité du typage de *Vier*, et les notions annexes. Compactifie la notation de la grammaire abstraite et définit le sucre syntaxique.
- révision**  $\beta_4$  Ajoute "Nothing".
- révision**  $\beta_3$  Modifie la syntaxe d'assignement des valeurs à un champ, qui permet maintenant une forme abrégée qui assigne plusieurs champs simultanément.
- révision**  $\beta_2$  Ajoute le paramètre manquant à "printInt" et "printChar" dans la grammaire abstraite et syntaxique.
- révision**  $\beta_1$  Première version publique.